
    Date: Thursday, October 10, 2019 @ 17:29:15
  Author: diabonas
Revision: 514487

archrelease: copy trunk to community-any

Added:
  pdftk/repos/community-any/
  pdftk/repos/community-any/PKGBUILD
    (from rev 514486, pdftk/trunk/PKGBUILD)
  pdftk/repos/community-any/pdftk.sh
    (from rev 514486, pdftk/trunk/pdftk.sh)

----------+
 PKGBUILD |   39 +++++++++++++++++++++++++++++++++++++++
 pdftk.sh |    3 +++
 2 files changed, 42 insertions(+)

Copied: pdftk/repos/community-any/PKGBUILD (from rev 514486, pdftk/trunk/PKGBUILD)
===================================================================
--- community-any/PKGBUILD	                        (rev 0)
+++ community-any/PKGBUILD	2019-10-10 17:29:15 UTC (rev 514487)
@@ -0,0 +1,39 @@
+# Maintainer: Jonas Witschel <diabonas at archlinux.org>
+pkgname=pdftk
+pkgver=3.0.7
+pkgrel=1
+pkgdesc='Command-line tool for working with PDFs'
+arch=('any')
+url='https://gitlab.com/pdftk-java/pdftk'
+license=('GPL')
+depends=('bcprov' 'java-commons-lang' 'java-runtime-headless')
+makedepends=('ant')
+checkdepends=('java-hamcrest' 'junit' 'junit-system-rules' 'poppler')
+source=("$url/-/archive/v$pkgver/$pkgname-v$pkgver.tar.bz2" 'pdftk.sh')
+sha512sums=('22e144a186b77817ff909543f068031023d7ebcc762443fc5ab4a31f008b9d62dd5662b7d122657e22e95292ea4b6443c4539ff279024d024aac92d7582b13f4'
+            '439c397472012d71c8e782a75819a7a950474c387b7050bbd5aa2f849d1dbe29423accf3f8e17d18436ba32eb4f76072df5b4f85caebd853df1e75c131631988')
+
+prepare() {
+	cd "$pkgname-v$pkgver"
+	mkdir lib
+	ln -s /usr/share/java/{bcprov,commons-lang/commons-lang,hamcrest-core,junit,junit-system-rules/system-rules}.jar lib
+	# No need for code coverage reports during tests
+	sed -ri '/<\/?jacoco:coverage>/d' build.xml
+}
+
+build() {
+	cd "$pkgname-v$pkgver"
+	ant jar
+}
+
+check() {
+	cd "$pkgname-v$pkgver"
+	ant test-run
+}
+
+package() {
+	cd "$pkgname-v$pkgver"
+	install -Dm644 build/jar/pdftk.jar -t "$pkgdir/usr/share/java/$pkgname"
+	install -Dm644 pdftk.1 -t "$pkgdir/usr/share/man/man1"
+	install -Dm755 "$srcdir/pdftk.sh" "$pkgdir/usr/bin/pdftk"
+}

Copied: pdftk/repos/community-any/pdftk.sh (from rev 514486, pdftk/trunk/pdftk.sh)
===================================================================
--- community-any/pdftk.sh	                        (rev 0)
+++ community-any/pdftk.sh	2019-10-10 17:29:15 UTC (rev 514487)
@@ -0,0 +1,3 @@
+#!/bin/sh
+CP='/usr/share/java/bcprov.jar:/usr/share/java/commons-lang/commons-lang.jar:/usr/share/java/pdftk/pdftk.jar'
+exec /usr/bin/java -cp "$CP" com.gitlab.pdftk_java.pdftk "$@"

